Description

`tpoisson` fits a truncated Poisson regression model when the number of occurrences of an event is restricted to be above a truncation point, below a truncation point, or between two truncation points. Truncated Poisson models are appropriate when neither the dependent variable nor the covariates are observed in the truncated part of the distribution. By default, `tpoisson` assumes left-truncation occurs at zero, but truncation may be specified at other fixed points or at values that vary across observations.

Quick start

Truncated Poisson regression of `y` on `x1` and `x2` with left-truncation at 0
`tpoisson y x1 x2`

Add categorical variable `a` using `factor-variable` syntax
`tpoisson y x1 x2 i.a`

As above, but report incidence-rate ratios and use a constant truncation point of 4
`tpoisson y x1 x2 i.a, irr ll(4)`

With offset variable `lnexp`
`tpoisson y x1 x2 i.a, offset(lnexp)`

As above, but with a variable truncation point stored in variable `min`
`tpoisson y x1 x2 i.a, offset(lnexp) ll(min)`

With variable left- and right-truncation points
`tpoisson y x1 x2, ll(min) ul(max)`

With variable right-truncation points
`tpoisson y x1 x2, ul(max)`

Constrain the coefficients for `2.a` and `3.a` to equality
`constraint define 1 2.a = 3.a`
`tpoisson y x1 x2 i.a, constraints(1)`

Menu

Statistics > Count outcomes > Truncated Poisson regression
Syntax

tpoisson  depvar  [ indepvars ]  [ if ]  [ in ]  [ weight ]  [ ,  options ]

options  Description

Model

noconstant  suppress constant term
ll(# | varname)  lower limit for truncation; default is 11(0) when neither
                ll() nor ul() is specified
ul(# | varname)  upper limit for truncation
exposure(varname)  include ln(varname) in model with coefficient constrained to 1
offset(varname)  include varname in model with coefficient constrained to 1
constraints(constraints)  apply specified linear constraints

SE/Robust

vce(vcetype)  vcetype may be oim, robust, cluster clustvar, opg, bootstrap, or jackknife

Reporting

level(#)  set confidence level; default is level(95)
irr  report incidence-rate ratios
noconsreport  do not display constraints
display_options  control columns and column formats, row spacing, line width,
                  display of omitted variables and base and empty cells, and
                  factor-variable labeling

Maximization

maximize_options  control the maximization process; seldom used

collinear  keep collinear variables
coefflegend  display legend instead of statistics

Options

noconstant; see [R] Estimation options.

ll(# | varname) and ul(# | varname) specify the lower and upper limits for truncation, respectively.
You may specify nonnegative integer values for one or both.
When neither ll() nor ul() is specified, the default is zero truncation, ll(0), equivalent to
left-truncation at zero.
tpoisson — Truncated Poisson regression

exposure(varname), offset(varname), constraints(constraints); see [R] Estimation options.

vce(vcetype) specifies the type of standard error reported, which includes types that are derived from asymptotic theory (oim, opg), that are robust to some kinds of misspecification (robust), that allow for intragroup correlation (cluster clustvar), and that use bootstrap or jackknife methods (bootstrap, jackknife); see [R] vce_option.

level(#); see [R] Estimation options.

irr reports estimated coefficients transformed to incidence-rate ratios, that is, $e^{\beta_i}$ rather than $\beta_i$. Standard errors and confidence intervals are similarly transformed. This option affects how results are displayed, not how they are estimated. irr may be specified at estimation or when replaying previously estimated results.

nocnsreport; see [R] Estimation options.

display_options: noci, nopvalues, noomitted, vsquish, noemptycells, baselevels, allbaselevels, nofvlabel, fvwrap(#), fvwrapon(style), cformat(%fmt), pformat(%fmt), sformat(%fmt), and nolstretch; see [R] Estimation options.

Maximization

maximize_options: difficult, technique(algorithm_spec), iterate(#), [no]log, trace, gradient, showstep, hessian, showtolerance, tolerance(#), ltolerance(#), nrtolerance(#), nonrtolerance, and from(init_specs); see [R] Maximize. These options are seldom used.

Setting the optimization type to technique(bhhh) resets the default vcetype to vce(opg).

The following options are available with tpoisson but are not shown in the dialog box: collinear, coeflegend; see [R] Estimation options.

Remarks and examples

tpoisson fits a truncated Poisson regression model by maximum likelihood estimation when the number of occurrences of an event is restricted to be above a truncation point, below a truncation point, or between two truncation points. If the dependent variable is not truncated, standard Poisson regression may be more appropriate; see [R] poisson.

When the data are truncated, we do not observe either the dependent variable or the covariates. For example, consider a study about the number of days that individuals with hyperglycemia are hospitalized after presenting to the hospital. If we select our sample only from admission records, then the sample is truncated at zero because we have data only on individuals who stayed at least one day. Now assume that we are relying on billing data and that hospitals may submit either a final bill when a patient is discharged or an interim bill every 30 days. In this case, we have no information about patients who are hospitalized fewer than 1 day or more than 30 days. Our data are left-truncated at 0 and right-truncated at 30.

A related phenomenon is censoring. For censored observations, we observe complete covariate information but only a censored value of the dependent variable. Different research designs can give rise to censored data or truncated data. See [R] cpoisson for information about censored Poisson regression.

Example 1: Left-truncation at zero

Consider the Simonoff (2003) dataset of running shoes for a sample of runners who registered an online running log. A running-shoe marketing executive is interested in knowing how the number of pairs of running shoes purchased relates to other factors such as gender, marital status, age, education, income, typical number of runs per week, average miles run per week, and the preferred type of running. These data are naturally truncated at zero. A truncated Poisson model is fit to the number of pairs of shoes owned on runs per week, miles run per week, gender, age, and marital status.

No options are needed because zero truncation is the default for tpoisson.

```
. use https://www.stata-press.com/data/r16/runshoes
. tpoisson shoes rpweek mpweek male age married
Iteration 0: log likelihood = -88.328151
Iteration 1: log likelihood = -86.272639
Iteration 2: log likelihood = -86.257999
Iteration 3: log likelihood = -86.257994
Truncated Poisson regression                         Number of obs = 60
limits: lower = 0                             LR chi2(5) = 22.75
          upper = +inf                         Prob > chi2 = 0.0004
Log likelihood = -86.257994                   Pseudo R2 = 0.1165

shoes | Coef.   Std. Err.     z    P>|z|     [95% Conf. Interval]
-------+---------------------------------------------
rpweek |  .1575811   .1097893  1.44   0.151    -.057602    .3727641
mpweek |  .0210673   .0091113  2.31   0.021     .0032094    .0389252
male   |   .0446134   .2444626  0.18   0.855    -.4345246    .5237513
age    |  .0185565   .0137786  1.35   0.178    -.008449    .0455622
married|  -.1283912   .2785044 -0.46   0.645    -.6742498    .4174674
_cons  |  -1.205844   .6619774 -1.82   0.069   -2.503296    .0916078
```

Using the zero-truncated Poisson regression with these data, only the coefficient on average miles per week is statistically significant at the 5% level.

Example 2: Left-truncation with a fixed-truncation point

Semiconductor manufacturing requires that silicon wafers be coated with a layer of metal oxide. The depth of this layer is strictly controlled. In this example, a critical oxide layer is designed for 300 ± 20 angstroms (Å).

After the oxide layer is coated onto a wafer, the wafer enters a photolithography step in which the lines representing the electrical connections are printed on the oxide and later etched and filled with metal. The widths of these lines are measured. In this example, they are controlled to 90±5 micrometers (µm).

After these and other steps, each wafer is electrically tested at probe. If too many failures are discovered, the wafer is rejected and sent for engineering analysis. In this example, the maximum number of probe failures tolerated for this product is 10.
A major failure at probe has been encountered—88 wafers had more than 10 failures each. The 88 wafers that failed were tested using 4 probe machines. The engineer suspects that the failures were a result of faulty probe machines, poor depth control, or poor line widths. The line widths and depths in these data are the actual measurement minus its specification target, 300 Å for the oxide depths and 90 µm for the line widths.

The following table tabulates the average failure rate for each probe using Stata’s mean command; see \[R\] mean.

```
. use https://www.stata-press.com/data/r16/probe
. mean failures, over(probe)

Mean estimation
Number of obs = 88

Mean     Std. Err.  [95% Conf. Interval]
c.failures@probe
1     15.8750   1.186293   13.51711   18.23289
2     14.9583   .591238   13.78318   16.13348
3     16.4706   .927987   14.62611   18.31506
4     23.0968   .945112   21.21826   24.97529
```

The 95% confidence intervals in this table suggest that there are about 5–11 additional failures per wafer on probe 4. These are unadjusted for varying line widths and oxide depths. Possibly, probe 4 received the wafers with larger line widths or extreme oxide depths.

Truncated Poisson regression more clearly identifies the root causes for the increased failures by estimating the differences between probes adjusted for the line widths and oxide depths. It also allows us to determine whether the deviations from specifications in line widths or oxide depths might be contributing to the problem.

```
. tpoisson failures i.probe depth width, ll(10) nolog

Truncated Poisson regression
Number of obs = 88
Limits: lower = 10
          LR chi2(5)  = 73.70
          upper = +inf
          Prob > chi2 = 0.0000
Log likelihood = -239.35746  Pseudo R2 = 0.1334

failures    Coef.  Std. Err.      z    P>|z|     [95% Conf. Interval]
probe
2    -.111304    .101979   -1.09   0.275    -.311178    .0885707
3     .011434    .103603    0.11   0.912    -.191625    .2144924
4     .425412    .084128    5.06   0.000     .260524    .5902989
depth   -.000503    .003338   -0.15   0.880    -.007045    .006038
width    .033023    .015573    2.12   0.034     .002501    .0635456
_cons   2.714025    .075262    36.06   0.000     2.566515    2.861536
```

The coefficients listed for the probes are testing the null hypothesis: \( H_0: \text{probe}_i = \text{probe}_1 \), where \( i \) equals 2, 3, and 4. Because the only coefficient that is statistically significant is the one for testing for \( H_0: \text{probe}_4 = \text{probe}_1 \), \( p < 0.001 \), and because the \( p \)-values for the other probes are not statistically significant, that is, \( p \geq 0.275 \), the implication is that there is a difference between probe 4 and the other machines. Because the coefficient for this test is positive, 0.425, the conclusion is that the average failure rate for probe 4, after adjusting for line widths and oxide depths, is higher than the other probes. Possibly, probe 4 needs calibration or the head used with this machine is defective.

Line-width control is statistically significant, \( p = 0.034 \), but variation in oxide depths is not causing the increased failure rate. The engineer concluded that the sudden increase in failures is the result of
two problems. First, probe 4 is malfunctioning, and second, there is a possible lithography or etching problem.

**Stored results**

tpoisson stores the following in e():

Scalars

<table>
<thead>
<tr>
<th>Scalar</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>e(N)</td>
<td>number of observations</td>
</tr>
<tr>
<td>e(k)</td>
<td>number of parameters</td>
</tr>
<tr>
<td>e(k_eq)</td>
<td>number of equations in e(b)</td>
</tr>
<tr>
<td>e(k_eq_model)</td>
<td>number of equations in overall model test</td>
</tr>
<tr>
<td>e(k_dv)</td>
<td>number of dependent variables</td>
</tr>
<tr>
<td>e(df_m)</td>
<td>model degrees of freedom</td>
</tr>
<tr>
<td>e(r2_p)</td>
<td>pseudo-$R^2$-squared</td>
</tr>
<tr>
<td>e(ll)</td>
<td>log likelihood</td>
</tr>
<tr>
<td>e(ll_0)</td>
<td>log likelihood, constant-only model</td>
</tr>
<tr>
<td>e(N_clust)</td>
<td>number of clusters</td>
</tr>
<tr>
<td>e(chi2)</td>
<td>$\chi^2$</td>
</tr>
<tr>
<td>e(p)</td>
<td>p-value for model test</td>
</tr>
<tr>
<td>e(rank)</td>
<td>rank of e(V)</td>
</tr>
<tr>
<td>e(ic)</td>
<td>number of iterations</td>
</tr>
<tr>
<td>e(rc)</td>
<td>return code</td>
</tr>
<tr>
<td>e(converged)</td>
<td>1 if converged, 0 otherwise</td>
</tr>
</tbody>
</table>

Macros

<table>
<thead>
<tr>
<th>Macro</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>e(cmd)</td>
<td>tpoisson</td>
</tr>
<tr>
<td>e(cmdline)</td>
<td>command as typed</td>
</tr>
<tr>
<td>e(depvar)</td>
<td>name of dependent variable</td>
</tr>
<tr>
<td>e(llopt)</td>
<td>contents of ll(), or 0 if neither ll() nor ul() is specified</td>
</tr>
<tr>
<td>e(ulopt)</td>
<td>contents of ul(), if specified</td>
</tr>
<tr>
<td>e(wtype)</td>
<td>weight type</td>
</tr>
<tr>
<td>e(wexp)</td>
<td>weight expression</td>
</tr>
<tr>
<td>e(title)</td>
<td>title in estimation output</td>
</tr>
<tr>
<td>e(clustvar)</td>
<td>name of cluster variable</td>
</tr>
<tr>
<td>e(offset)</td>
<td>linear offset variable</td>
</tr>
<tr>
<td>e(chi2type)</td>
<td>Wald or LR: type of model $\chi^2$ test</td>
</tr>
<tr>
<td>e(vce)</td>
<td>vcetype specified in vce()</td>
</tr>
<tr>
<td>e(vcetype)</td>
<td>title used to label Std. Err.</td>
</tr>
<tr>
<td>e(opt)</td>
<td>type of optimization</td>
</tr>
<tr>
<td>e(which)</td>
<td>max or min; whether optimizer is to perform maximization or minimization</td>
</tr>
<tr>
<td>e(ml_method)</td>
<td>type of ml method</td>
</tr>
<tr>
<td>e(user)</td>
<td>name of likelihood-evaluator program</td>
</tr>
<tr>
<td>e(technique)</td>
<td>maximization technique</td>
</tr>
<tr>
<td>e(properties)</td>
<td>b V</td>
</tr>
<tr>
<td>e(predict)</td>
<td>program used to implement predict</td>
</tr>
<tr>
<td>e(asbalanced)</td>
<td>factor variables fvset as asbalanced</td>
</tr>
<tr>
<td>e(asobserved)</td>
<td>factor variables fvset as asobserved</td>
</tr>
</tbody>
</table>

Matrices

<table>
<thead>
<tr>
<th>Matrix</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>e(b)</td>
<td>coefficient vector</td>
</tr>
<tr>
<td>e(Cns)</td>
<td>constraints matrix</td>
</tr>
<tr>
<td>e(ilog)</td>
<td>iteration log (up to 20 iterations)</td>
</tr>
<tr>
<td>e(gradient)</td>
<td>gradient vector</td>
</tr>
<tr>
<td>e(V)</td>
<td>variance–covariance matrix of the estimators</td>
</tr>
<tr>
<td>e(V_modelbased)</td>
<td>model-based variance</td>
</tr>
</tbody>
</table>

Functions

<table>
<thead>
<tr>
<th>Function</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>e(sample)</td>
<td>marks estimation sample</td>
</tr>
</tbody>
</table>
Methods and formulas

For a nonnegative count outcome $Y$ with left-truncation point $ll_j$ and right-truncation point $ul_j$, we can write the truncated Poisson model as

$$f(y_j) = \frac{\exp(-\lambda_j)\lambda_j^{y_j}}{y_j!Pr(ll_j < Y < ul_j | \xi_j)}$$

where

$$\xi_j = x_j\beta + \text{offset}_j$$

$$\lambda_j = \exp(\xi_j)$$

and $x_j$ is a vector of observed covariates. The conditional probability of observing $y_j$ events, therefore given by $ll_j < y_j < ul_j$, is

$$\Pr(Y = y_j | ll_j < y_j < ul_j, x_j) = \frac{\exp(-\lambda_j)\lambda_j^{y_j}}{y_j!Pr(ll_j < Y < ul_j | x_j)}$$

The log likelihood is given by

$$\ln L = \sum_{j=1}^{n} w_j \left[ -\lambda_j + \xi_j y_j - \ln(y_j!) - \ln \{ \Pr(ll_j < Y < ul_j | \xi_j) \} \right]$$

If no weights are specified, $w_j = 1$.

This command supports the Huber/White/sandwich estimator of the variance and its clustered version using \texttt{vce(robust)} and \texttt{vce(cluster clustvar)}, respectively. See \texttt{[P] \_robust}, particularly \texttt{Maximum likelihood estimators} and \texttt{Methods and formulas}.

tpoisson also supports estimation with survey data. For details on VCEs with survey data, see \texttt{[SVY] Variance estimation}.

Acknowledgment

We gratefully acknowledge the previous work by Joseph Hilbe (1944–2017) (1999), a former editor of the \textit{Stata Technical Bulletin} and coauthor of the Stata Press book \textit{Generalized Linear Models and Extensions}.

References


——. 2010. \textit{Microeconometrics Using Stata}. Rev. ed. College Station, TX: Stata Press.


**Also see**

[R] tpoisson postestimation — Postestimation tools for tpoisson

[R] poisson — Poisson regression

[R] nbreg — Negative binomial regression

[R] tnbreg — Truncated negative binomial regression

[R] zinb — Zero-inflated negative binomial regression

[R] zip — Zero-inflated Poisson regression

[BAYES] bayes: tpoisson — Bayesian truncated Poisson regression

[FMM] fmm: tpoisson — Finite mixtures of truncated Poisson regression models

[SVY] svy estimation — Estimation commands for survey data

[XT] xtpoisson — Fixed-effects, random-effects, and population-averaged Poisson models

[U] 20 Estimation and postestimation commands